

# 9/7

## Jeux d'Arcade

---

Ces jeux ont remplacé le « baby-foot » de nos jeunes années. Seul le flipper semble survivre au raz-de-marée de ces nouveaux venus.

Transformez votre Amstrad en une véritable console de jeu.

Utilisez, améliorez les programmes de ce chapitre et envoyez-nous vos œuvres originales.

### 9/7.1

## Casse briques

---

Le classique jeu de café est repris dans ce programme. Un mur de briques de 4 briques d'épaisseur occupe la partie haute de l'écran. Une raquette peut être déplacée en utilisant les touches \ et Z. Vous disposez de 5 balles pour arriver à un score minimum de 1 300 points. Votre score est affiché en fin de partie.

```

1000 'Casse briques
1010 '=====
1020 GOSUB 1060 'Initialisation du jeu
1030 GOSUB 2150 'fin de la partie
1040 END
1050 '=====
1060 'Initialisation
1070 '=====
1080 MODE 1:ON BREAK GOSUB 2210
1090 FOR I=0 TO 6:READ A:POKE &9000+I,A:NEXT I
1100 DATA &CD,&60,&BB,&32,&10,&90,&C9
1110 SPEED KEY 1,1
1120 '-----
1130 'Definition des briques et de la raquette
1140 '-----
1150 SYMBOL AFTER 129
1160 SYMBOL 129,73,54,73,54,73,54,73,54
1170 SYMBOL 130,0,0,0,0,0,0,255,255,0
1180 '-----
1190 'Affichage des briques
1200 '-----
1210 CLS
1220 FOR I=1 TO 4
1230   FOR K=1 TO 38
1240     PRINT CHR$(129);
1250     NEXT K
1260   PRINT
1270 NEXT I
1280 '-----
1290 'Affichage de la raquette
1300 '-----
1310 XR=1:YR=20
1320 LOCATE XR,YR:PRINT CHR$(130);CHR$(130);CHR$(130)
1330 '-----
1340 XB=3:YB=19
1350 VX=INT(RND(1)*3)-1:VY=-1
1360 '-----
1370 'Debut de la partie
1380 '-----
1390 PRINT"           Balle 1":BA=1:PRINT
1400 PRINT"Appuyez sur une touche pour commencer"
1410 A#=INKEY#:IF A#="" THEN 1410
1420 LOCATE 1,23:PRINT SPACE$(38)
1430 PRINT"           SCORE : ";SC
1440 '-----
1450 'Deroulement du jeu
1460 '-----
1470 GOSUB 1500
1480 B#=INKEY#:IF B#<>"" THEN GOSUB 1850
1490 IF FIN=1 THEN RETURN ELSE 1470
1500 '-----
1510 'Action sur la balle
1520 '-----
1530 LOCATE XB,YB:PRINT" "
1540 IF (YB+VY)<=0 THEN VY=1:VX=INT(RND(1)*3)-1
1550 IF XB+VX<1 THEN VX=1:SOUND 1,100,10
1560 IF XB+VX=39 THEN VX=-1:SOUND 1,100,10
1570 LOCATE XB+VX,YB+VY:CALL &9000:A=PEEK(&9010):IF A=129 THEN GOSUB 1710
1580 IF A=130 THEN GOSUB 1650
1590 IF YB+VY>20 THEN GOSUB 2040
1600 IF FIN=1 THEN RETURN
1610 IF XB+VX<1 THEN VX=1:SOUND 1,100,10
1620 IF XB+VX=39 THEN VX=-1:SOUND 1,100,10

```

```

1630 XB=XB+VX:YB=YB+VY:LOCATE XB,YB:PRINT"o"
1640 RETURN
1650 '-----
1660 'Renvoi de la balle
1670 '-----
1680 VY=-1:VX=INT(RND(1)*3)-1
1690 SOUND 1,200,10
1700 RETURN
1710 '-----
1720 'Demolition brique
1730 '-----
1740 LOCATE XB+VX,YB+VY:PRINT" "
1750 SC=SC+10*(5-YB-VY):LOCATE 14,24:PRINT"SCORE : ";SC
1760 VY=1:VX=INT(RND(1)*3)-1
1770 IF SC>1300 THEN FIN=1
1780 SOUND 1,100,10
1790 RETURN
1800 '-----
1810 'Balle perdue
1820 '-----
1830 FIN=1
1840 RETURN
1850 '-----
1860 'Action du joueur
1870 '-----
1880 IF UPPER$(B$)="Z" THEN GOSUB 1910
1890 IF B$="c" THEN GOSUB 1970
1900 RETURN
1910 '-----
1920 'Deplacement de la raquette <-
1930 '-----
1940 IF XR=1 THEN RETURN
1950 LOCATE XR+1,YR:PRINT" ":XR=XR-2:LOCATE XR,YR:PRINT CHR$(130);CHR$(130)
1960 RETURN
1970 '-----
1980 'Deplacement de la raquette ->
1990 '-----
2000 IF XR=37 THEN RETURN
2010 LOCATE XR,YR:PRINT" ":XR=XR+2:LOCATE XR+1,YR:PRINT CHR$(130);CHR$(130)
2020 RETURN
2030 '-----
2040 REM Perte de la balle
2050 '-----
2060 BA=BA+1:IF BA=6 THEN FIN=1:GOTO 2130 ELSE FIN=0
2070 LOCATE 1,21:PRINT" Balle";BA:PRINT
2080 LOCATE 1,23:PRINT"Appuyez sur une touche pour commencer"
2090 A$=INKEY$:IF A$="" THEN 2090
2100 LOCATE 1,23:PRINT SPACE$(38);
2110 XB=3:YB=19:VY=-1
2120 PRINT:PRINT:PRINT
2130 RETURN
2140 '-----
2150 REM Fin du jeu
2160 '-----
2170 CLS
2180 PRINT"Score";SC
2190 IF BA=1 OR BA=2 THEN PRINT "Bravo"
2200 SPEED KEY 10,1
2210 RETURN

```

Lignes 1000 à 1040 : Programme principal

Lignes 1050 à 1350 : Initialisation (définition des caractères graphiques, et affichage du terrain de jeu)

Lignes 1360 à 1430 : Début d'une partie

Lignes 1440 à 1490 : Déroulement d'une partie

Lignes 1500 à 1840 : Déplacement de la balle en fonction des obstacles rencontrés

Lignes 1850 à 2020 : Action du joueur

Lignes 2030 à 2130 : Prise en compte de la perte d'une balle

Lignes 2140 à 2210 : Fin du jeu

Un programme en assembleur est inséré dans le listing pour permettre de lire le code d'un caractère n'importe où sur l'écran. Pour cela, nous faisons appel à la macro du firmware « TXD RD CHAR » qui a son point d'entrée en &BB60. Consultez la Partie 4 Chap. 2.7 pour avoir plus de détails sur le fonctionnement de cette routine. Le code ASCII du caractère est placé en &9010, et lu en BASIC par l'ordre PEEK.

```
1          ORG 9000H
2          LOAD 9000H
3 9000 CD60BB          CALL 0BB60H          ;Input char
4 9003 321090         LD (9010H),A          ;Sauvegarde lecture
5 9006 C9            RET
6                  END
```